Creators: J. de Graaf and S. Samin

Created: 24.04.2019

Version: Mathematica 11.2.0.0

In this Mathematica notebook we solve the doubly linearized equation system governing self-thermo(di)electrophoresis for the intermediate regime of ionic strength. We also consider the two limiting cases, obtaining analytic expressions that we report on in the main text.

Introducing Useful Functions

In the following we will make use of spherical polar coordinates. It is therefore useful to introduce the gradient, divergence, and Laplacian for this system.

```
In[1]:= GradSPH[fun_] := D[fun, r] * {1, 0} + (1 / r) * D[fun, θ] * {0, 1}
        DivSPH[fvec_] := (1/ (r^2)) * D[(r^2) * fvec[[1]], r] + (1/ (r * Sin[θ])) * D[Sin[θ] * fvec[[2]], θ]
        LapSPH[fun_] := DivSPH[GradSPH[fun]];

In[4]:= (* Check the forms of the derivatives. *)

In[5]:= Simplify[GradSPH[f[r, θ]]]
        Simplify[DivSPH[{R[r, θ], T[r, θ]}]]
        Simplify[LapSPH[f[r, θ]]]
```

Out[5]= $\left\{f^{(1,0)}[r,\theta], \frac{f^{(0,1)}[r,\theta]}{r}\right\}$

Out[6]= $\frac{2\,R[r,\theta] + \mathrm{Cot}[\theta]\,T[r,\theta] + T^{(0,1)}[r,\theta] + r\,R^{(1,0)}[r,\theta]}{r}$

Out[7]= $\frac{\mathrm{Cot}[\theta]\,f^{(0,1)}[r,\theta] + f^{(0,2)}[r,\theta] + 2\,r\,f^{(1,0)}[r,\theta] + r^2\,f^{(2,0)}[r,\theta]}{r^2}$

The Equilibrium Equations

Here we solve for the reduced equilibrium electrostatic potential, which we will need in constructing and solving equations for the non-equilibrium fields. Throughout we will assume a homogeneous boundary condition for the electrostatic potential.

```
In[8]:= (* The equation for the equilibrium electrostatic potential. *)

In[9]:= phidiff = (LapSPH[pheq[r]] == κ^2 * pheq[r]);
        phieq = Expand[Simplify[DSolve[phidiff, pheq[r], r], Assumptions → {κ > 0}][[1, 1, 2]]]
```

Out[10]= $\frac{e^{-r\kappa}\,C[1]}{r} + \frac{e^{r\kappa}\,C[2]}{2\,r\,\kappa}$

```
In[11]:= (* The part that diverges can be eliminated. *)
```

In[12]:=
```
phieqn = phieq /. {C[2] → 0}
```
Out[12]= $\frac{e^{-r\kappa} C[1]}{r}$

In[13]:=
```
(* The electrostatic potential for a conducting surface. *)
```

In[14]:=
```
subcond = Solve[(phieqn /. {r → a}) == ϕ0, C[1]][[1]];
ϕceq[r_] := phieqn /. subcond;
ϕceq[r]
```
Out[16]= $\frac{a\, e^{a\kappa - r\kappa}\, \phi 0}{r}$

In[17]:=
```
(* The electrostatic potential for an insulating surface. *)
```

In[18]:=
```
dphieqn = Simplify[GradSPH[phieqn]][[1]];
subiso = Solve[(dphieqn /. {r → a}) == -σ0 * el / (ϵ0 * kB * T0), C[1]][[1]];
ϕseq[r_] := phieqn /. subiso;
ϕseq[r]
```
Out[21]= $\frac{a^2\, e^{a\kappa - r\kappa}\, el\, \sigma 0}{kB\, r\, T0\, \epsilon 0\, (1 + a\kappa)}$

In[22]:=
```
(* Note that these two only differ in terms of the prefactor. Hence and for convenience we will assume the generic form. *)
```

In[23]:=
```
ϕeq[r_, θ_] := ϕs * (a / r) * Exp[-κ * r];
ϕeq[r, θ]
```
Out[24]= $\frac{a\, e^{-r\kappa}\, \phi s}{r}$

The Non-Equilibrium Equations: The Temperature

The temperature field is easily solved for, because it is a simple solution to the Laplace equation.

In[25]:=
```
(* Assume the following form for the temperature field. *)
```

In[26]:=
```
T[r_, θ_] := tr[r] * Cos[θ];
```

In[27]:=
```
(* Specify the differential equation. *)
```

In[28]:=
```
tdiff = Simplify[LapSPH[T[r, θ]] == 0, Assumptions → {r > 0, Cos[θ] > 0}]
```
Out[28]= $2\, tr[r] == r\, (2\, tr'[r] + r\, tr''[r])$

In[29]:=
```
(* Solve for this equation. *)
```

In[30]:=
```
tdsol = DSolve[tdiff, tr[r], r][[1, 1, 2]]
```
Out[30]= $r\, C[1] + \frac{C[2]}{r^2}$

In[31]:=
```
(* So that the full solution is given by. *)
```

In[32]:=
```
tds = tdsol * Cos[θ]
```
Out[32]= $\left(r\, C[1] + \frac{C[2]}{r^2}\right) Cos[\theta]$

In[33]:=
```
(* Now we must study the limit to infinity of this equation. *)
```

In[34]:=
```
Limit[tds, r → Infinity]
```
Out[34]= $C[1]\,Cos[\theta]\,\infty$

In[35]:=
```
(* Clearly we must drop the C[1] prefactor term. *)
```
In[36]:=
```
tdsn = tds /. {C[1] → 0}
```
Out[36]= $\frac{C[2]\,Cos[\theta]}{r^2}$

In[37]:=
```
Limit[tdsn, r → Infinity]
```
Out[37]= 0

In[38]:=
```
(* The integration constant can be set by using the boundary condition. *)
```
In[39]:=
```
tsubs = Solve[Simplify[(tdsn /. {r → a}) == Cos[θ]], C[2]][[1]]
```
Out[39]= $\{C[2] \to a^2\}$

In[40]:=
```
(* Solution to the reduced heat equation. *)
```
In[41]:=
```
tdsn /. tsubs
```
Out[41]= $\frac{a^2\,Cos[\theta]}{r^2}$

In[42]:=
```
(* Note that we have to play a trick to get the symbols recognised properly by
 Mathematica. Also we introduce a minus sign to account for the hot cap being on the bottom (negative values of z). *)
```
In[43]:=
```
t[p_, q_] := -(tdsn /. tsubs /. {r → p, θ → q});
t[r, θ]
```
Out[44]= $-\frac{a^2\,Cos[\theta]}{r^2}$

The Non-Equilibrium Equations: The Ionic Strength

Next we take the steps necessary to solve for the non-equilibrium fields. We start with the local ionic strength which is decoupled from the other equations, other than through the boundary conditions.

In[45]:=
```
(* Assume the following form for the local non-equilibrium ionic strength *)
```
In[46]:=
```
X[r_, θ_] := x[r] * Cos[θ];
```
In[47]:=
```
(* Then the associated differential equation becomes. *)
```
In[48]:=
```
diffXlhs = Simplify[LapSPH[X[r, θ]]]
```
Out[48]= $\frac{Cos[\theta]\,\left(-2\,x[r] + r\,\left(2\,x'[r] + r\,x''[r]\right)\right)}{r^2}$

In[49]:=
```
diffXrhs = Simplify[β * Dot[GradSPH[t[r, θ]], GradSPH[ϕeq[r, θ]]]]
```
Out[49]= $-\frac{2\,a^3\,e^{-r\,\kappa}\,\beta\,(1 + r\,\kappa)\,\phi s\,Cos[\theta]}{r^5}$

In[50]:=
```
(* With this the piece of interest. *)
```
In[51]:=
```
xdiff = FullSimplify[diffXlhs == diffXrhs, Assumptions → {r > 0, a > 0, Cos[θ] > 0}]
```
Out[51]= $2\,a^3\,\beta\,(1 + r\,\kappa)\,\phi s + e^{r\,\kappa}\,r^3\,\left(-2\,x[r] + r\,\left(2\,x'[r] + r\,x''[r]\right)\right) == 0$

```
In[52]:= (* Now solve this differential equation. *)

In[53]:= xdsol = DSolve[xdiff, x[r], r][[1, 1, 2]]
```

Out[53]= $$r\,C[1] + \frac{C[2]}{r^2} + \frac{a^3\, e^{-r\kappa}\, \beta\, \phi s\, \left(-6 + 2\, r\, \kappa - r^2\, \kappa^2 + r^3\, \kappa^3 + e^{r\kappa}\, r^4\, \kappa^4\, \mathrm{ExpIntegralEi}[-r\,\kappa]\right)}{12\, r^3}$$

```
In[54]:= (* So that the full solution is given by. *)

In[55]:= xds = Simplify[Expand[xdsol * Cos[θ]]]
```

Out[55]= $$\frac{\mathrm{Cos}[\theta]\, \left(a^3\, e^{-r\kappa}\, \beta\, \left(-6 + 2\, r\, \kappa - r^2\, \kappa^2 + r^3\, \kappa^3\right)\, \phi s + 12\, r\, \left(r^3\, C[1] + C[2]\right) + a^3\, r^4\, \beta\, \kappa^4\, \phi s\, \mathrm{ExpIntegralEi}[-r\,\kappa]\right)}{12\, r^3}$$

```
In[56]:= (* Now we must study the limit to infinity of this equation. *)

In[57]:= Limit[xds, r → Infinity, Assumptions → {a > 0, κ > 0}]
```

Out[57]= $C[1]\,\mathrm{Cos}[\theta]\,\infty$

```
In[58]:= (* We conclude that we must drop the C[1] prefactored term. *)

In[59]:= xdsn = xds /. {C[1] → 0}
```

Out[59]= $$\frac{\mathrm{Cos}[\theta]\, \left(a^3\, e^{-r\kappa}\, \beta\, \left(-6 + 2\, r\, \kappa - r^2\, \kappa^2 + r^3\, \kappa^3\right)\, \phi s + 12\, r\, C[2] + a^3\, r^4\, \beta\, \kappa^4\, \phi s\, \mathrm{ExpIntegralEi}[-r\,\kappa]\right)}{12\, r^3}$$

```
In[60]:= Limit[xdsn, r → Infinity, Assumptions → {a > 0, κ > 0}]
```

Out[60]= 0

```
In[61]:= (* We cannot as of yet determine the integration constant so that we arrive at. *)

In[62]:= Xs[p_, q_] := (xdsn /. {C[2] → Xint, r → p, θ → q});
         Xs[r, θ]
```

Out[63]= $$\frac{\mathrm{Cos}[\theta]\, \left(12\, r\, \mathrm{Xint} + a^3\, e^{-r\kappa}\, \beta\, \left(-6 + 2\, r\, \kappa - r^2\, \kappa^2 + r^3\, \kappa^3\right)\, \phi s + a^3\, r^4\, \beta\, \kappa^4\, \phi s\, \mathrm{ExpIntegralEi}[-r\,\kappa]\right)}{12\, r^3}$$

The Non-Equilibrium Equations: The Ion Excess

As we argue in the main text, the local ion excess for may be isolated from the non-equilibrium electrostatic potential, which makes it an ideal target to solve for next.

```
In[64]:= (* Assume the following form for the local non-equilibrium ionic strength. *)

In[65]:= dX[r_, θ_] := dx[r] * Cos[θ];

In[66]:= (* Then the associated differential equation becomes. *)

In[67]:= diffdXlhs = Simplify[Expand[LapSPH[dX[r, θ]] - (κ^2) * dX[r, θ]]]
```

Out[67]= $$\frac{\mathrm{Cos}[\theta]\, \left(-\left(2 + r^2\, \kappa^2\right)\, dx[r] + r\, \left(2\, dx'[r] + r\, dx''[r]\right)\right)}{r^2}$$

```
In[68]:= diffdXrhs = Simplify[(1 + γ + ϵst) * Dot[GradSPH[t[r, θ]], GradSPH[ϕeq[r, θ]]] + (κ^2) * (1 + ϵst) * t[r, θ] * ϕeq[r, θ]]
```

Out[68]= $$-\frac{a^3\, e^{-r\kappa}\, \left(2\, \gamma\, (1 + r\, \kappa) + (1 + \epsilon st)\, \left(2 + 2\, r\, \kappa + r^2\, \kappa^2\right)\right)\, \phi s\, \mathrm{Cos}[\theta]}{r^5}$$

```
In[69]:= (* With this the piece of interest. *)
```

In[70]:=
```
dxdiff = FullSimplify[diffdXlhs == diffdXrhs, Assumptions → {r > 0, a > 0, Cos[θ] > 0}]
```
Out[70]=
$$a^3 \left(2\,\gamma\,(1+r\,\kappa) + (1+\epsilon st)\left(2+r\,\kappa\,(2+r\,\kappa)\right)\right)\phi s = e^{r\,\kappa}\,r^3 \left(\left(2+r^2\,\kappa^2\right) dx[r] - r\left(2\,dx'[r] + r\,dx''[r]\right)\right)$$

In[71]:=
```
(* Now solve this differential equation. *)
```

In[72]:=
```
dxdsol = DSolve[dxdiff, dx[r], r][[1, 1, 2]];
sdxdsol = Simplify[dxdsol, Assumptions → {a > 0, κ > 0, r > 0}]
```
Out[73]=
$$\frac{1}{4\,r^4\,\kappa} e^{-2\,r\,\kappa}\left(-\left(a^3\left(\gamma+2\,r\,\gamma\,\kappa+(1+\epsilon st)\,(1+r\,\kappa)^2+e^{2\,r\,\kappa}\left(\gamma\left(-1+2\,r^2\,\kappa^2\right)+(1+\epsilon st)\left(-1+r^2\,\kappa^2+2\,r^3\,\kappa^3\right)\right)\right)\phi s-4\,e^{2\,r\,\kappa}\,r^4\,\kappa\,C[1]\right)\text{SphericalBesselJ}[-2,\, i\,r\,\kappa] - i\left(a^3\left(-(1+\epsilon st)\,(1+r\,\kappa)^2-\gamma\left(1+2\,r\,\kappa\right)+e^{2\,r\,\kappa}\left(\gamma\left(-1+2\,r^2\,\kappa^2\right)+(1+\epsilon st)\left(-1+r^2\,\kappa^2+2\,r^3\,\kappa^3\right)\right)\right)\phi s-4\,i\,e^{2\,r\,\kappa}\,r^4\,\kappa\,C[2]\right)\text{SphericalBesselY}[-2,\, i\,r\,\kappa]\right)$$

In[74]:=
```
(* So that the full solution is given by. *)
```

In[75]:=
```
dxds = Simplify[Expand[sdxdsol * Cos[θ]]]
```
Out[75]=
$$-\frac{1}{4\,r^4\,\kappa} e^{-2\,r\,\kappa}\,\text{Cos}[\theta]\left(\left(a^3\left(\gamma+2\,r\,\gamma\,\kappa+(1+\epsilon st)\,(1+r\,\kappa)^2+e^{2\,r\,\kappa}\left(\gamma\left(-1+2\,r^2\,\kappa^2\right)+(1+\epsilon st)\left(-1+r^2\,\kappa^2+2\,r^3\,\kappa^3\right)\right)\right)\phi s-4\,e^{2\,r\,\kappa}\,r^4\,\kappa\,C[1]\right)\text{SphericalBesselJ}[-2,\, i\,r\,\kappa] + \left(i\,a^3\left(-(1+\epsilon st)\,(1+r\,\kappa)^2-\gamma\left(1+2\,r\,\kappa\right)+e^{2\,r\,\kappa}\left(\gamma\left(-1+2\,r^2\,\kappa^2\right)+(1+\epsilon st)\left(-1+r^2\,\kappa^2+2\,r^3\,\kappa^3\right)\right)\right)\phi s+4\,e^{2\,r\,\kappa}\,r^4\,\kappa\,C[2]\right)\text{SphericalBesselY}[-2,\, i\,r\,\kappa]\right)$$

In[76]:=
```
(* Now we must study the limit to infinity of this equation. *)
```

In[77]:=
```
Limit[dxds, r → Infinity, Assumptions → {a > 0, κ > 0}]
```
Out[77]=
$$\left(C[1] - i\,C[2]\right)\text{Cos}[\theta]\,(-\infty)$$

In[78]:=
```
(* So that we must choose C[1] such that this term drops. *)
```

In[79]:=
```
dxdsn = FullSimplify[dxds /. {C[1] → I * C[2]}, Assumptions → {a > 0, κ > 0, r > 0}]
```
Out[79]=
$$-\frac{e^{-r\,\kappa}\left(a^3\,\kappa^2\,(1+\gamma+\epsilon st+r\,(1+\epsilon st)\,\kappa)\,\phi s-2\,i\,r\,(1+r\,\kappa)\,C[2]\right)\text{Cos}[\theta]}{2\,r^3\,\kappa^2}$$

In[80]:=
```
Limit[dxdsn, r → Infinity, Assumptions → {a > 0, κ > 0}]
```
Out[80]=
$$0$$

In[81]:=
```
(* Note that the above implies that we are left with complex values in our expression. This means that C[2] must be some complex number as well. This is
   surprising from a physics perspective. Hence we fix this in assigning the as-of-yet-unknown integration constant. We are always free to make this choice. *)
```

In[82]:=
```
dXs[p_, q_] := (dxdsn /. {C[2] → I * dXint, r → p, θ → q});
dXs[r, θ]
```
Out[83]=
$$-\frac{e^{-r\,\kappa}\left(2\,dXint\,r\,(1+r\,\kappa)+a^3\,\kappa^2\,(1+\gamma+\epsilon st+r\,(1+\epsilon st)\,\kappa)\,\phi s\right)\text{Cos}[\theta]}{2\,r^3\,\kappa^2}$$

The Non-Equilibrium Equations: The Poisson Equation

Now that we have a solution for the local ion excess, we can solve for the non-equilibrium part of the Poisson Equation, which depends on this quantity.

In[84]:=
```
(* Assume the following form for the local non-equilibrium ionic strength. *)
```

In[85]:=
```
P[r_, θ_] := p[r] * Cos[θ];
```

In[86]:=
```
(* Then the associated differential equation becomes. *)
```

In[87]:=
```
diffPlhs = Simplify[LapSPH[P[r, θ]]]
```
Out[87]=
$$\frac{\text{Cos}[\theta]\left(-2\,p[r]+r\left(2\,p'[r]+r\,p''[r]\right)\right)}{r^2}$$

In[88]:=
```
diffPrhs = Simplify[-εst * DivSPH[t[r, θ] * GradSPH[ϕeq[r, θ]]] - (κ^2) * dXs[r, θ]]
```

Out[88]=
$$\frac{e^{-r\kappa}\left(2\,\text{dXint}\,r^3(1+r\kappa)+a^3\left(r^2\kappa^2(1+\gamma+r\kappa)+\epsilon\text{st}\left(4+4r\kappa+3r^2\kappa^2+r^3\kappa^3\right)\right)\phi\text{s}\right)\text{Cos}[\theta]}{2r^5}$$

In[89]:=
```
(* With this the piece of interest. *)
```

In[90]:=
```
pdiff = FullSimplify[diffPlhs == diffPrhs, Assumptions → {r > 0, a > 0, κ > 0, Cos[θ] > 0}]
```

Out[90]=
$$2\,\text{dXint}\,r^3(1+r\kappa)+a^3\left(r^2\kappa^2(1+\gamma+r\kappa)+\epsilon\text{st}(2+r\kappa)(2+r\kappa(1+r\kappa))\right)\phi\text{s} == 2e^{r\kappa}r^3\left(-2p[r]+r\left(2p'[r]+r\,p''[r]\right)\right)$$

In[91]:=
```
(* Now solve this differential equation. *)
```

In[92]:=
```
pdsol = DSolve[pdiff, p[r], r][[1, 1, 2]];
spdsol = Simplify[pdsol, Assumptions → {a > 0, κ > 0, r > 0}]
```

Out[93]=
$$\frac{1}{12r^3\kappa^2}e^{-r\kappa}\left(12\,\text{dXint}\,r(1+r\kappa)+\kappa^2\left(a^3\left(6\epsilon\text{st}(1+r\kappa)+r\kappa\left(4+r\kappa-r^2\kappa^2+\gamma\left(2-r\kappa+r^2\kappa^2\right)\right)\right)\phi\text{s}+12e^{r\kappa}r\left(r^3C[1]+C[2]\right)\right)+a^3e^{r\kappa}r^4(-1+\gamma)\kappa^6\phi\text{s}\,\text{ExpIntegralEi}[-r\kappa]\right)$$

In[94]:=
```
(* So that the full solution is given by. *)
```

In[95]:=
```
pds = Simplify[Expand[spdsol * Cos[θ]]]
```

Out[95]=
$$\frac{1}{12r^3\kappa^2}e^{-r\kappa}\text{Cos}[\theta]\left(12\,\text{dXint}\,r(1+r\kappa)+\kappa^2\left(a^3\left(6\epsilon\text{st}(1+r\kappa)+r\kappa\left(4+r\kappa-r^2\kappa^2+\gamma\left(2-r\kappa+r^2\kappa^2\right)\right)\right)\phi\text{s}+12e^{r\kappa}r\left(r^3C[1]+C[2]\right)\right)+a^3e^{r\kappa}r^4(-1+\gamma)\kappa^6\phi\text{s}\,\text{ExpIntegralEi}[-r\kappa]\right)$$

In[96]:=
```
(* Now we must study the limit to infinity of this equation. *)
```

In[97]:=
```
Limit[pds, r → Infinity, Assumptions → {a > 0, κ > 0}]
```

Out[97]=
$$C[1]\,\text{Cos}[\theta]\,\infty$$

In[98]:=
```
(* Clearly we must drop the C[1] prefactor. *)
```

In[99]:=
```
pdsn = FullSimplify[pds /. {C[1] → 0}, Assumptions → {a > 0, κ > 0, r > 0}]
```

Out[99]=
$$\frac{1}{12r^3\kappa^2}e^{-r\kappa}\text{Cos}[\theta]\left(12\,\text{dXint}\,r(1+r\kappa)+\kappa^2\left(a^3\left(2r(2+\gamma)\kappa+r^2(-1+\gamma)\kappa^2(-1+r\kappa)+6\epsilon\text{st}(1+r\kappa)\right)\phi\text{s}+12e^{r\kappa}r\,C[2]\right)+a^3e^{r\kappa}r^4(-1+\gamma)\kappa^6\phi\text{s}\,\text{ExpIntegralEi}[-r\kappa]\right)$$

In[100]:=
```
Limit[pdsn, r → Infinity, Assumptions → {a > 0, κ > 0}]
```

Out[100]=
$$0$$

In[101]:=
```
(* We must still determine the integration constant. *)
```

In[102]:=
```
Ps[p_, q_] := (pdsn /. {C[2] → Pint, r → p, θ → q});
Ps[r, θ]
```

Out[103]=
$$\frac{1}{12r^3\kappa^2}e^{-r\kappa}\text{Cos}[\theta]\left(12\,\text{dXint}\,r(1+r\kappa)+\kappa^2\left(12e^{r\kappa}\text{Pint}\,r+a^3\left(2r(2+\gamma)\kappa+r^2(-1+\gamma)\kappa^2(-1+r\kappa)+6\epsilon\text{st}(1+r\kappa)\right)\phi\text{s}\right)+a^3e^{r\kappa}r^4(-1+\gamma)\kappa^6\phi\text{s}\,\text{ExpIntegralEi}[-r\kappa]\right)$$

The Non-Equilibrium Equations: Fixing the Integration Constants

We now have to use the flux boundary conditions on the particle for X and dX to fix the integration constants. The integration constant for the potential must be specified separately for conducting and insulating surfaces. Here we use a two-vector notation, where the first component corresponds to the radial contribution and the second to the polar.

In[104]:=
```
(* The flux of positive ions is given by. *)
```

In[105]:=
```
fluxp = Simplify[(β + γ) * (ϕeq[r, θ] - 1) * GradSPH[t[r, θ]] - GradSPH[Xs[r, θ]] - GradSPH[dXs[r, θ]] + t[r, θ] * GradSPH[ϕeq[r, θ]] - GradSPH[Ps[r, θ]]];
fluxp // MatrixForm
```

Out[106]//MatrixForm=

$$\begin{pmatrix} \frac{1}{12}\text{Cos}[\theta]\left(\frac{24\,(\text{Pint}+\text{Xint}-a^2(\beta+\gamma))}{r^3}-\frac{a^3 e^{-r\kappa}(-1+\beta+\gamma)(-6+2r\kappa-r^2\kappa^2+r^3\kappa^3)\,\phi s}{r^4}-a^3(-1+\beta+\gamma)\kappa^4\phi s\,\text{ExpIntegralEi}[-r\kappa]\right) \\ \frac{1}{12}\left(\frac{12\,(\text{Pint}+\text{Xint}-a^2(\beta+\gamma))}{r^3}+\frac{a^3 e^{-r\kappa}(-1+\beta+\gamma)(6+2r\kappa-r^2\kappa^2+r^3\kappa^3)\,\phi s}{r^4}+a^3(-1+\beta+\gamma)\kappa^4\phi s\,\text{ExpIntegralEi}[-r\kappa]\right)\text{Sin}[\theta] \end{pmatrix}$$

In[107]:=
```
(* The flux of negative ions is given by. *)
```

In[108]:=
```
fluxm = Simplify[(β - γ) * (ϕeq[r, θ] + 1) * GradSPH[t[r, θ]] - GradSPH[Xs[r, θ]] + GradSPH[dXs[r, θ]] - t[r, θ] * GradSPH[ϕeq[r, θ]] + GradSPH[Ps[r, θ]]];
fluxm // MatrixForm
```

Out[109]//MatrixForm=

$$\begin{pmatrix} \frac{1}{12}\text{Cos}[\theta]\left(\frac{24\,(-\text{Pint}+\text{Xint}+a^2(\beta-\gamma))}{r^3}+\frac{a^3 e^{-r\kappa}(1+\beta-\gamma)(6-2r\kappa+r^2\kappa^2-r^3\kappa^3)\,\phi s}{r^4}-a^3(1+\beta-\gamma)\kappa^4\phi s\,\text{ExpIntegralEi}[-r\kappa]\right) \\ \frac{1}{12}\left(\frac{12\,(-\text{Pint}+\text{Xint}+a^2(\beta-\gamma))}{r^3}+\frac{a^3 e^{-r\kappa}(1+\beta-\gamma)(6+2r\kappa-r^2\kappa^2+r^3\kappa^3)\,\phi s}{r^4}+a^3(1+\beta-\gamma)\kappa^4\phi s\,\text{ExpIntegralEi}[-r\kappa]\right)\text{Sin}[\theta] \end{pmatrix}$$

In[110]:=
```
(* Now evaluate these at the surface dotted with the normal to the surface *)
```

In[111]:=
```
fpa = Dot[fluxp, {1, 0}] /. {r → a}
fma = Dot[fluxm, {1, 0}] /. {r → a}
```

Out[111]=

$$\frac{1}{12}\text{Cos}[\theta]\left(\frac{24\,(\text{Pint}+\text{Xint}-a^2(\beta+\gamma))}{a^3}-\frac{e^{-a\kappa}(-1+\beta+\gamma)(-6+2a\kappa-a^2\kappa^2+a^3\kappa^3)\,\phi s}{a}-a^3(-1+\beta+\gamma)\kappa^4\phi s\,\text{ExpIntegralEi}[-a\kappa]\right)$$

Out[112]=

$$\frac{1}{12}\text{Cos}[\theta]\left(\frac{24\,(-\text{Pint}+\text{Xint}+a^2(\beta-\gamma))}{a^3}+\frac{e^{-a\kappa}(1+\beta-\gamma)(6-2a\kappa+a^2\kappa^2-a^3\kappa^3)\,\phi s}{a}-a^3(1+\beta-\gamma)\kappa^4\phi s\,\text{ExpIntegralEi}[-a\kappa]\right)$$

In[113]:=
```
(* Intriguingly this expression does not depend on dXint. One might have expected such a dependence as the fluxes only depend on X and dX and hence their
 integration constants. However this is an expression of the interconnectedness of dX and ϕ. We proceed by solving for zero flux through the boundary. *)
```

In[114]:=
```
fsub = Solve[{fpa == 0, fma == 0}, {Pint, Xint}][[1]];
fsub // MatrixForm
```

Out[115]//MatrixForm=

$$\begin{pmatrix} \text{Pint}\to\frac{1}{24}a^2 e^{-a\kappa}\left(24e^{a\kappa}\beta+6\phi s-6\gamma\phi s-2a\kappa\phi s+2a\gamma\kappa\phi s+a^2\kappa^2\phi s-a^2\gamma\kappa^2\phi s-a^3\kappa^3\phi s+a^3\gamma\kappa^3\phi s-a^4e^{a\kappa}\kappa^4\phi s\,\text{ExpIntegralEi}[-a\kappa]+a^4e^{a\kappa}\gamma\kappa^4\phi s\,\text{ExpIntegralEi}[-a\kappa]\right) \\ \text{Xint}\to\frac{1}{24}a^2e^{-a\kappa}\left(24e^{a\kappa}\gamma-6\beta\phi s+2a\beta\kappa\phi s-a^2\beta\kappa^2\phi s+a^3\beta\kappa^3\phi s+a^4e^{a\kappa}\beta\kappa^4\phi s\,\text{ExpIntegralEi}[-a\kappa]\right) \end{pmatrix}$$

In[116]:=
```
(* Now we substitute this solution into the original expressions. *)
```

In[117]:=
```
Print["The Local Non-Equilibrium Ionic Strength:"]
Xsol = Simplify[Expand[Xs[r, θ] /. fsub]]
Print["The Local Non-Equilibrium Ion Excess:"]
dXsol = Simplify[Expand[dXs[r, θ] /. fsub]]
Print["The Non-Equilibrium Potential:"]
Psol = Simplify[Expand[Ps[r, θ] /. fsub]]
```

The Local Non-Equilibrium Ionic Strength:

Out[118]=

$$\frac{1}{24 r^3} a^2 e^{-(a+r)\kappa} \mathrm{Cos}[\theta] \Big(24 e^{(a+r)\kappa} r \gamma - 12 a e^{a\kappa} \beta \phi s - 6 e^{r\kappa} r \beta \phi s + 4 a e^{a\kappa} r \beta \kappa \phi s + 2 a e^{r\kappa} r \beta \kappa \phi s - a^2 e^{r\kappa} r \beta \kappa^2 \phi s - 2 a e^{a\kappa} r^2 \beta \kappa^2 \phi s + a^3 e^{r\kappa} r \beta \kappa^3 \phi s + 2 a e^{a\kappa} r^3 \beta \kappa^3 \phi s + a^4 e^{(a+r)\kappa} r \beta \kappa^4 \phi s \, \mathrm{ExpIntegralEi}[-a\kappa] + 2 a e^{(a+r)\kappa} r^4 \beta \kappa^4 \phi s \, \mathrm{ExpIntegralEi}[-r\kappa]\Big)$$

The Local Non-Equilibrium Ion Excess:

Out[120]=

$$-\frac{e^{-r\kappa} \big(2\, \mathrm{dXint}\, r (1 + r\kappa) + a^3 \kappa^2 (\gamma + (1 + \epsilon st)(1 + r\kappa)) \phi s\big) \mathrm{Cos}[\theta]}{2 r^3 \kappa^2}$$

The Non-Equilibrium Potential:

Out[122]=

$$\frac{1}{24 r^3 \kappa^2} e^{-(a+r)\kappa} \mathrm{Cos}[\theta] \Big(24\, \mathrm{dXint}\, e^{a\kappa} r + 24\, \mathrm{dXint}\, e^{a\kappa} r^2 \kappa + 24 a^2 e^{(a+r)\kappa} r \beta \kappa^2 + 6 a^2 e^{r\kappa} r \kappa^2 \phi s - 6 a^2 e^{r\kappa} r \gamma \kappa^2 \phi s + 12 a^3 e^{a\kappa} \epsilon st\, \kappa^2 \phi s + 8 a^3 e^{a\kappa} r \kappa^3 \phi s - 2 a^3 e^{r\kappa} r \kappa^3 \phi s + 4 a^3 e^{a\kappa} r \gamma \kappa^3 \phi s + 2 a^3 e^{r\kappa} r \gamma \kappa^3 \phi s + 12 a^3 e^{a\kappa} r \epsilon st\, \kappa^3 \phi s + a^4 e^{r\kappa} r \kappa^4 \phi s + 2 a^3 e^{a\kappa} r^2 \kappa^4 \phi s - a^4 e^{r\kappa} r \gamma \kappa^4 \phi s - 2 a^3 e^{a\kappa} r^2 \gamma \kappa^4 \phi s - a^5 e^{r\kappa} r \kappa^5 \phi s - 2 a^3 e^{a\kappa} r^3 \kappa^5 \phi s + a^5 e^{r\kappa} r \gamma \kappa^5 \phi s + 2 a^3 e^{a\kappa} r^3 \gamma \kappa^5 \phi s + a^6 e^{(a+r)\kappa} r (-1 + \gamma) \kappa^6 \phi s \, \mathrm{ExpIntegralEi}[-a\kappa] + 2 a^3 e^{(a+r)\kappa} r^4 (-1 + \gamma) \kappa^6 \phi s \, \mathrm{ExpIntegralEi}[-r\kappa]\Big)$$

In[123]:=
```
(* Next we must determine the remaining integration constant dXint from the electrostatic boundary conditions. We do this for a conducting surface first by equating the
 potential to zero at the surface. Note that the equilibrium electrostatic potential already accounted for the entirety of the electrostatic boundary condition. *)
```

In[124]:=
```
Print["The Integration Constant for a Conducting Surface:"]
solcond = Solve[Simplify[Psol //. {r → a}] == 0, dXint][[1]]
Print["The Local Non-Equilibrium Ionic Strength:"]
Xctot = Simplify[Expand[Xsol]]
Print["The Local Non-Equilibrium Ion Excess:"]
dXctot = Simplify[Expand[dXsol /. solcond]]
Print["The Non-Equilibrium Potential:"]
Pctot = Simplify[Expand[Psol /. solcond]]
```

The Integration Constant for a Conducting Surface:

Out[125]=
$$\left\{\mathrm{dXint} \to -\frac{1}{8\,(1+a\,\kappa)} a^2\,\kappa^2 \left(8\,e^{a\,\kappa}\,\beta + 2\,\phi s - 2\,\gamma\,\phi s + 4\,\epsilon st\,\phi s + 2\,a\,\kappa\,\phi s + 2\,a\,\gamma\,\kappa\,\phi s + 4\,a\,\epsilon st\,\kappa\,\phi s + a^2\,\kappa^2\,\phi s - a^2\,\gamma\,\kappa^2\,\phi s - a^3\,\kappa^3\,\phi s + a^3\,\gamma\,\kappa^3\,\phi s - a^4\,e^{a\,\kappa}\,\kappa^4\,\phi s\,\mathrm{ExpIntegralEi}[-a\,\kappa] + a^4\,e^{a\,\kappa}\,\gamma\,\kappa^4\,\phi s\,\mathrm{ExpIntegralEi}[-a\,\kappa]\right)\right\}$$

The Local Non-Equilibrium Ionic Strength:

Out[127]=
$$\frac{1}{24\,r^3} a^2\,e^{-(a+r)\,\kappa}\,\mathrm{Cos}[\theta] \left(24\,e^{(a+r)\,\kappa}\,r\,\gamma - 12\,a\,e^{a\,\kappa}\,\beta\,\phi s - 6\,e^{r\,\kappa}\,r\,\beta\,\phi s + 4\,a\,e^{a\,\kappa}\,r\,\beta\,\kappa\,\phi s + 2\,a\,e^{r\,\kappa}\,r\,\beta\,\kappa\,\phi s - a^2\,e^{r\,\kappa}\,r\,\beta\,\kappa^2\,\phi s - 2\,a\,e^{a\,\kappa}\,r^2\,\beta\,\kappa^2\,\phi s + a^3\,e^{r\,\kappa}\,r\,\beta\,\kappa^3\,\phi s + 2\,a\,e^{a\,\kappa}\,r^3\,\beta\,\kappa^3\,\phi s + a^4\,e^{(a+r)\,\kappa}\,r\,\beta\,\kappa^4\,\phi s\,\mathrm{ExpIntegralEi}[-a\,\kappa] + 2\,a\,e^{(a+r)\,\kappa}\,r^4\,\beta\,\kappa^4\,\phi s\,\mathrm{ExpIntegralEi}[-r\,\kappa]\right)$$

The Local Non-Equilibrium Ion Excess:

Out[129]=
$$\frac{1}{8\,r^3\,(1+a\,\kappa)} a^2\,e^{-r\,\kappa}\,\mathrm{Cos}[\theta] \left(8\,e^{a\,\kappa}\,r\,\beta\,(1+r\,\kappa) + \left(-2\,r\,(-1+\gamma-2\,\epsilon st)\,(1+r\,\kappa) + a^3\,r\,(-1+\gamma)\,\kappa^3\,(1+r\,\kappa) + 2\,a\left((1+r\,\kappa)\,(-2-2\,\epsilon st+r\,\kappa+2\,r\,\epsilon st\,\kappa) + \gamma\,(-2+r\,\kappa+r^2\,\kappa^2)\right) - a^2\,\kappa\left((4+4\,\epsilon st-r\,\kappa)\,(1+r\,\kappa) + \gamma\,(4+r\,\kappa+r^2\,\kappa^2)\right)\right)\phi s + a^4\,e^{a\,\kappa}\,r\,(-1+\gamma)\,\kappa^4\,(1+r\,\kappa)\,\phi s\,\mathrm{ExpIntegralEi}[-a\,\kappa]\right)$$

The Non-Equilibrium Potential:

Out[131]=
$$\frac{1}{24\,r^3\,(1+a\,\kappa)} a^2\,e^{-(a+r)\,\kappa}\,\mathrm{Cos}[\theta] \left(-24\,e^{2\,a\,\kappa}\,r\,\beta + 24\,e^{(a+r)\,\kappa}\,r\,\beta + 24\,a\,e^{(a+r)\,\kappa}\,r\,\beta\,\kappa - 24\,e^{2\,a\,\kappa}\,r^2\,\beta\,\kappa - 6\,e^{a\,\kappa}\,r\,\phi s + 6\,e^{r\,\kappa}\,r\,\phi s + 6\,e^{a\,\kappa}\,r\,\gamma\,\phi s - 6\,e^{r\,\kappa}\,r\,\gamma\,\phi s + 12\,a\,e^{a\,\kappa}\,\epsilon st\,\phi s - 12\,e^{a\,\kappa}\,r\,\epsilon st\,\phi s + 2\,a\,e^{a\,\kappa}\,r\,\kappa\,\phi s + 4\,a\,e^{r\,\kappa}\,r\,\kappa\,\phi s - 6\,e^{a\,\kappa}\,r^2\,\kappa\,\phi s - 2\,a\,e^{a\,\kappa}\,r\,\gamma\,\kappa\,\phi s - 4\,a\,e^{r\,\kappa}\,r\,\gamma\,\kappa\,\phi s + 6\,e^{a\,\kappa}\,r^2\,\gamma\,\kappa\,\phi s + 12\,a^2\,e^{a\,\kappa}\,\epsilon st\,\kappa\,\phi s - 12\,e^{a\,\kappa}\,r^2\,\epsilon st\,\kappa\,\phi s + 5\,a^2\,e^{a\,\kappa}\,r\,\kappa^2\,\phi s - a^2\,e^{r\,\kappa}\,r\,\kappa^2\,\phi s - 4\,a\,e^{a\,\kappa}\,r^2\,\kappa^2\,\phi s + 7\,a^2\,e^{a\,\kappa}\,r\,\gamma\,\kappa^2\,\phi s + a^2\,e^{r\,\kappa}\,r\,\gamma\,\kappa^2\,\phi s - 8\,a\,e^{a\,\kappa}\,r^2\,\gamma\,\kappa^2\,\phi s + 12\,a^2\,e^{a\,\kappa}\,r\,\epsilon st\,\kappa^2\,\phi s - 12\,a\,e^{a\,\kappa}\,r^2\,\epsilon st\,\kappa^2\,\phi s + 3\,a^3\,e^{a\,\kappa}\,r\,\kappa^3\,\phi s - a^2\,e^{a\,\kappa}\,r^2\,\kappa^3\,\phi s - 2\,a\,e^{a\,\kappa}\,r^3\,\kappa^3\,\phi s - 3\,a^3\,e^{a\,\kappa}\,r\,\gamma\,\kappa^3\,\phi s + a^2\,e^{a\,\kappa}\,r^2\,\gamma\,\kappa^3\,\phi s + 2\,a\,e^{a\,\kappa}\,r^3\,\gamma\,\kappa^3\,\phi s - a^4\,e^{r\,\kappa}\,r\,\kappa^4\,\phi s + 3\,a^3\,e^{a\,\kappa}\,r^2\,\kappa^4\,\phi s - 2\,a^2\,e^{a\,\kappa}\,r^3\,\kappa^4\,\phi s + a^4\,e^{r\,\kappa}\,r\,\gamma\,\kappa^4\,\phi s - 3\,a^3\,e^{a\,\kappa}\,r^2\,\gamma\,\kappa^4\,\phi s + 2\,a^2\,e^{a\,\kappa}\,r^3\,\gamma\,\kappa^4\,\phi s + a^4\,r\,(-1+\gamma)\,\kappa^4\left(e^{(a+r)\,\kappa}\,(1+a\,\kappa) - 3\,e^{2\,a\,\kappa}\,(1+r\,\kappa)\right)\phi s\,\mathrm{ExpIntegralEi}[-a\,\kappa] + 2\,a\,e^{(a+r)\,\kappa}\,r^4\,(-1+\gamma)\,\kappa^4\,(1+a\,\kappa)\,\phi s\,\mathrm{ExpIntegralEi}[-r\,\kappa]\right)$$

In[132]:=
```
(* For the insulating surface we must set the gradient of the potential equal to zero to ensure that no surface charge is added. *)
```

In[133]:=
```
Print["The Integration Constant for an Insulating Surface:"]
soliso = Solve[Simplify[(Dot[GradSPH[Psol], {1, 0}]) //. {r → a}] == 0, dXint][[1]]
Print["The Local Non-Equilibrium Ionic Strength:"]
Xstot = Xctot
Print["The Local Non-Equilibrium Ion Excess:"]
dXstot = Simplify[Expand[dXsol /. soliso]]
Print["The Non-Equilibrium Potential:"]
Pstot = Simplify[Expand[Psol /. soliso]]
```

```
The Integration Constant for an Insulating Surface:
```

Out[134]=

$$\left\{dXint \to -\frac{a^2 \kappa^2 \left(4\, \mathrm{e}^{a \kappa} \beta + \phi s - \gamma\, \phi s + 3\, \epsilon st\, \phi s + a \kappa\, \phi s + a \gamma \kappa\, \phi s + 3\, a\, \epsilon st\, \kappa\, \phi s + a^2 \kappa^2\, \phi s + a^2\, \epsilon st\, \kappa^2\, \phi s\right)}{2 \left(2 + 2 a \kappa + a^2 \kappa^2\right)}\right\}$$

```
The Local Non-Equilibrium Ionic Strength:
```

Out[136]=

$$\frac{1}{24 r^3} a^2 \mathrm{e}^{-(a+r)\kappa} \mathrm{Cos}[\theta] \Big(24\, \mathrm{e}^{(a+r)\kappa} r \gamma - 12\, a\, \mathrm{e}^{a\kappa} \beta\, \phi s - 6\, \mathrm{e}^{r\kappa} r \beta\, \phi s + 4\, a\, \mathrm{e}^{a\kappa} r \beta \kappa\, \phi s + 2\, a\, \mathrm{e}^{r\kappa} r \beta \kappa\, \phi s - a^2 \mathrm{e}^{r\kappa} r \beta \kappa^2\, \phi s - 2\, a\, \mathrm{e}^{a\kappa} r^2 \beta \kappa^2\, \phi s + a^3 \mathrm{e}^{r\kappa} r \beta \kappa^3\, \phi s + 2\, a\, \mathrm{e}^{a\kappa} r^3 \beta \kappa^3\, \phi s + a^4 \mathrm{e}^{(a+r)\kappa} r \beta \kappa^4\, \phi s\, \mathrm{ExpIntegralEi}[-a\kappa] + 2\, a\, \mathrm{e}^{(a+r)\kappa} r^4 \beta \kappa^4\, \phi s\, \mathrm{ExpIntegralEi}[-r\kappa]\Big)$$

```
The Local Non-Equilibrium Ion Excess:
```

Out[138]=

$$-\frac{1}{2 r^3 \left(2 + 2 a \kappa + a^2 \kappa^2\right)} a^2 \mathrm{e}^{-r\kappa} \Big(-4\, \mathrm{e}^{a\kappa} r \beta\, (1 + r\kappa) + \Big(r \left(-1 + \gamma - 3\, \epsilon st\right) (1 + r\kappa) + a^3 \kappa^2 \left(\gamma + (1 + \epsilon st)(1 + r\kappa)\right) + a^2 \kappa \left(2\gamma - (1 + \epsilon st)\left(-2 - r\kappa + r^2\kappa^2\right)\right) - a\left((1 + r\kappa)\left(-2 - 2\, \epsilon st + r\kappa + 3\, r\, \epsilon st\, \kappa\right) + \gamma\left(-2 + r\kappa + r^2\kappa^2\right)\right)\Big)\phi s\Big)\, \mathrm{Cos}[\theta]$$

```
The Non-Equilibrium Potential:
```

Out[140]=

$$\frac{1}{24 r^3 \left(2 + 2 a \kappa + a^2 \kappa^2\right)} a^2 \mathrm{e}^{-(a+r)\kappa} \mathrm{Cos}[\theta] \Big(-48\, \mathrm{e}^{2 a\kappa} r \beta + 48\, \mathrm{e}^{(a+r)\kappa} r \beta + 48\, a\, \mathrm{e}^{(a+r)\kappa} r \beta \kappa - 48\, \mathrm{e}^{2 a\kappa} r^2 \beta \kappa + 24\, a^2 \mathrm{e}^{(a+r)\kappa} r \beta \kappa^2 - 12\, \mathrm{e}^{a\kappa} r\, \phi s + 12\, \mathrm{e}^{r\kappa} r\, \phi s + 12\, \mathrm{e}^{a\kappa} r \gamma\, \phi s - 12\, \mathrm{e}^{r\kappa} r \gamma\, \phi s + 24\, a\, \mathrm{e}^{a\kappa} \epsilon st\, \phi s - 36\, \mathrm{e}^{a\kappa} r\, \epsilon st\, \phi s + 4\, a\, \mathrm{e}^{a\kappa} r \kappa\, \phi s +$$
$$8\, a\, \mathrm{e}^{r\kappa} r \kappa\, \phi s - 12\, \mathrm{e}^{a\kappa} r^2 \kappa\, \phi s - 4\, a\, \mathrm{e}^{a\kappa} r \gamma \kappa\, \phi s - 8\, a\, \mathrm{e}^{r\kappa} r \gamma \kappa\, \phi s + 12\, \mathrm{e}^{a\kappa} r^2 \gamma \kappa\, \phi s + 24\, a^2 \mathrm{e}^{a\kappa} \epsilon st\, \kappa\, \phi s - 12\, a\, \mathrm{e}^{a\kappa} r\, \epsilon st\, \kappa\, \phi s - 36\, \mathrm{e}^{a\kappa} r^2 \epsilon st\, \kappa\, \phi s + 4\, a^2 \mathrm{e}^{a\kappa} r \kappa^2\, \phi s + 4\, a^2 \mathrm{e}^{r\kappa} r \kappa^2\, \phi s - 8\, a\, \mathrm{e}^{a\kappa} r^2 \kappa^2\, \phi s +$$
$$8\, a^2 \mathrm{e}^{a\kappa} r \gamma \kappa^2\, \phi s - 4\, a^2 \mathrm{e}^{r\kappa} r \gamma \kappa^2\, \phi s - 16\, a\, \mathrm{e}^{a\kappa} r^2 \gamma \kappa^2\, \phi s + 12\, a^3 \mathrm{e}^{a\kappa} \epsilon st\, \kappa^2\, \phi s + 12\, a^2 \mathrm{e}^{a\kappa} r\, \epsilon st\, \kappa^2\, \phi s - 36\, a\, \mathrm{e}^{a\kappa} r^2 \epsilon st\, \kappa^2\, \phi s + 8\, a^3 \mathrm{e}^{a\kappa} r \kappa^3\, \phi s - 2\, a^3 \mathrm{e}^{r\kappa} r \kappa^3\, \phi s - 8\, a^2 \mathrm{e}^{a\kappa} r^2 \kappa^3\, \phi s - 4\, a\, \mathrm{e}^{a\kappa} r^3 \kappa^3\, \phi s + 4\, a^3 \mathrm{e}^{a\kappa} r \gamma \kappa^3\, \phi s +$$
$$2\, a^3 \mathrm{e}^{r\kappa} r \gamma \kappa^3\, \phi s - 4\, a^2 \mathrm{e}^{a\kappa} r^2 \gamma \kappa^3\, \phi s + 4\, a\, \mathrm{e}^{a\kappa} r^3 \gamma \kappa^3\, \phi s + 12\, a^3 \mathrm{e}^{a\kappa} r\, \epsilon st\, \kappa^3\, \phi s - 12\, a^2 \mathrm{e}^{a\kappa} r^2 \epsilon st\, \kappa^3\, \phi s - a^4 \mathrm{e}^{r\kappa} r \kappa^4\, \phi s + 2\, a^3 \mathrm{e}^{a\kappa} r^2 \kappa^4\, \phi s - 4\, a^2 \mathrm{e}^{a\kappa} r^3 \kappa^4\, \phi s + a^4 \mathrm{e}^{r\kappa} r \gamma \kappa^4\, \phi s - 2\, a^3 \mathrm{e}^{a\kappa} r^2 \gamma \kappa^4\, \phi s + 4\, a^2 \mathrm{e}^{a\kappa} r^3 \gamma \kappa^4\, \phi s -$$
$$a^5 \mathrm{e}^{r\kappa} r \kappa^5\, \phi s - 2\, a^3 \mathrm{e}^{a\kappa} r^3 \kappa^5\, \phi s + a^5 \mathrm{e}^{r\kappa} r \gamma \kappa^5\, \phi s + 2\, a^3 \mathrm{e}^{a\kappa} r^3 \gamma \kappa^5\, \phi s + a^4 \mathrm{e}^{(a+r)\kappa} r\, (-1 + \gamma)\, \kappa^4 \left(2 + 2 a \kappa + a^2 \kappa^2\right) \phi s\, \mathrm{ExpIntegralEi}[-a\kappa] + 2\, a\, \mathrm{e}^{(a+r)\kappa} r^4 (-1 + \gamma)\, \kappa^4 \left(2 + 2 a \kappa + a^2 \kappa^2\right) \phi s\, \mathrm{ExpIntegralEi}[-r\kappa]\Big)$$

The Swim Speed using Teubner’s Formalism

We now have solutions for the non-equilibrium potential and ion excess, as well as for the temperature and equilibrium potential, we can determine the speed of the particle.

In[141]:=
```
(* We first determine the force acting on the fluid. Here we need to distinguish between conducting and onsulating surfaces. For
 convenience we also strip off the dielectrophoretic contribution which is equal in both cases barring the obvious difference in ϕs. *)
```

In[142]:=
```
peq = ϕeq[r, θ];
temp = t[r, θ];
Fc = -2 * kB * T0 * n0 * (dXctot * GradSPH[peq] - peq * GradSPH[Pctot]);
Fs = -2 * kB * T0 * n0 * (dXstot * GradSPH[peq] - peq * GradSPH[Pstot]);
Fϵ = -2 * kB * T0 * n0 * ((1 / (2 * (κ^2))) * ϵst * Dot[GradSPH[peq], GradSPH[peq]] * GradSPH[temp]);
```

In[147]:=
```
(* The integration kernel for Teubner's formalism is given by. Here we also introduce a prefactor that we will use
 as a multiplicative constant towards the end of the calculation. We also incorporate the azimuthal integration into this prefactor. *)
```

In[148]:=
```
Kpref = (1 / (6 * Pi * eta * a)) * 2 * Pi;
Tker = {((3 * a) / (2 * r) - ((a^3) / (2 * (r^3))) - 1) * Cos[θ], -((3 * a) / (4 * r) + ((a^3) / (4 * (r^3))) - 1) * Sin[θ]};
Tker // MatrixForm
```

Out[150]//MatrixForm=

$$\begin{pmatrix} \left(-1-\frac{a^3}{2 r^3}+\frac{3 a}{2 r}\right) \mathrm{Cos}[\theta] \\ \left(1-\frac{a^3}{4 r^3}-\frac{3 a}{4 r}\right) \mathrm{Sin}[\theta] \end{pmatrix}$$

In[151]:=
```
(* So that the integrand is given by the following expressions. Here we also take into account the Jacobian of the transformation to
 spherical polar coordinates. Lastly the variables are exchanged again to ensure all terms are recognized by Mathematica's integration routine. *)
```

In[152]:=
```
Intc = (Dot[Tker, Fc] * (r^2) * Sin[θ]) //. {r → ri, θ → ti};
Ints = (Dot[Tker, Fs] * (r^2) * Sin[θ]) //. {r → ri, θ → ti};
Intϵ = (Dot[Tker, Fϵ] * (r^2) * Sin[θ]) //. {r → ri, θ → ti};
```

In[155]:=
```
(* Unfortunately the above presents a set of extremely challenging intgrands. We therefore break these up into terms and integrate the terms separately. *)
```

In[156]:=
```
exIntc = Expand[Intc];
Length[exIntc]
```

Out[157]= 413

In[158]:=
```
exInts = Expand[Ints];
Length[exInts]
```

Out[159]= 451

In[160]:=
```
exIntϵ = Expand[Intϵ];
Length[exIntϵ]
```

Out[161]= 18

In[162]:=
```
(* The routine to integrate a single term *)
```

In[163]:=
```
IntTerm[IIin_, Lin_] := Module[{II = IIin, L = Lin, p0, p0a, p1, xi, yi, res},
   p0 = Simplify[L[[II]]];
   p0a = Simplify[p0 //. {ri → a}];
   p1 = Simplify[p0 / p0a];
   yi = Simplify[Integrate[p0a, {ti, 0, Pi}]];
   xi = Integrate[p1, {ri, a, Infinity}, Assumptions → {a > 0, κ > 0}];
   res = Simplify[xi * yi];
   res];
```

The Speed resulting from the Dielectrophoretic Force Contribution

Let us consider the speed generated by the force term that scales with ϵ^* first. Note that through the dependence of the other fields on this parameter, there will be other contributions, but this term is relatively short and will give us a feeling for the way the integration routines perform. There are dielectric contributions to the other integrands as well. Thus, even knowing these terms here, does not give us sufficient information to analyse the Hückel and Smoluchowski limits.

In[164]:=
```
(*** Sum the full series for the dielectrophoretic term ***)
```

In[165]:=
```
(* Uncomment the following lines to perform the calculation *)
(*exSϵ=Sum[IntTerm[k,exIntϵ],{k,1,Length[exIntϵ]}];*)
(*SϵR=Kpref*Simplify[Expand[exSϵ]]*)
```

In[166]:=
```
(* The calculations are long so we present the result below. *)
```

In[167]:=

$$S\epsilon = \frac{1}{630\, a\, \text{eta}\, \kappa^2} e^{-2 a \kappa}\, \text{kB n0 T0}\, \epsilon\text{st}\, \phi\text{s}^2 \left(-12 - 24 a \kappa + 4 a^2 \kappa^2 + 5 a^3 \kappa^3 - 22 a^4 \kappa^4 - 6 a^5 \kappa^5 + 12 a^6 \kappa^6 + 8 a^5 e^{2 a \kappa} \kappa^5 \left(-7 + 3 a^2 \kappa^2\right) \text{ExpIntegralEi}[-2 a \kappa]\right);$$

In[168]:=
```
(* Dielectrophoretic term for conducting surface *)
```

In[169]:=
```
Sϵc = Simplify[Sϵ //. {ϕs → ϕ0 * Exp[a * κ], n0 → κ^2 * ϵ0 * kB * T0 / (2 * el^2)}]
```

Out[169]=

$$\frac{1}{1260\, a\, \text{el}^2\, \text{eta}} \text{kB}^2\, \text{T0}^2\, \epsilon 0\, \epsilon\text{st}\, \phi 0^2 \left(-12 - 24 a \kappa + 4 a^2 \kappa^2 + 5 a^3 \kappa^3 - 22 a^4 \kappa^4 - 6 a^5 \kappa^5 + 12 a^6 \kappa^6 + 8 a^5 e^{2 a \kappa} \kappa^5 \left(-7 + 3 a^2 \kappa^2\right) \text{ExpIntegralEi}[-2 a \kappa]\right)$$

In[170]:=
```
(* Dielectrophoretic term for conducting surface *)
```

In[171]:=
```
Sϵs = Simplify[Sϵ //. {ϕs → a * el * σ0 * Exp[a * κ] / (kB * T0 * ϵ0 * (1 + κ * a)), n0 → κ^2 * ϵ0 * kB * T0 / (2 * el^2)}]
```

Out[171]=

$$\frac{1}{1260\, \text{eta}\, \epsilon 0\, (1 + a \kappa)^2} a\, \epsilon\text{st}\, \sigma 0^2 \left(-12 - 24 a \kappa + 4 a^2 \kappa^2 + 5 a^3 \kappa^3 - 22 a^4 \kappa^4 - 6 a^5 \kappa^5 + 12 a^6 \kappa^6 + 8 a^5 e^{2 a \kappa} \kappa^5 \left(-7 + 3 a^2 \kappa^2\right) \text{ExpIntegralEi}[-2 a \kappa]\right)$$

The Speed resulting from the Other Forces for a Conducting Surface

Now that we have covered the dielectrophoretic force contribution, we turn our attention to the more complicated force expressions for the conductor. Here we add the dielectrophoretic contribution.

In[172]:=
```
(*** Sum the full series for the conducting surface ***)
```

In[173]:=
```
(* Uncomment the following lines to perform the calculation *)
(*exSc= Sum[IntTerm[k,exIntc],{k,1,Length[exIntc]}];*)
(*ScR=Kpref*Simplify[Expand[exSc]]*)
```

In[174]:=
```
(* The calculations are long so we present the result below. *)
```

In[175]:=
```
Sc =
 Simplify[(-1/(4320 eta (1 + a κ)) a e^(-2 a κ) kB n0 T0 ϕs (720 e^(a κ) β - 480 a e^(a κ) β κ - 1320 a^2 e^(a κ) β κ^2 + 120 a^4 e^(a κ) β κ^4 + 84 ϕs - 276 γ ϕs - 228 a κ ϕs + 372 a γ κ ϕs - 1204 a^2 κ^2 ϕs + 676 a^2 γ κ^2 ϕs - 900 a^3 κ^3 ϕs - 60 a^3 γ κ^3 ϕs - 37
     a^4 κ^4 ϕs + 133 a^4 γ κ^4 ϕs - 19 a^5 κ^5 ϕs + 211 a^5 γ κ^5 ϕs + 5 a^6 κ^6 ϕs - 5 a^6 γ κ^6 ϕs - 5 a^7 κ^7 ϕs + 5 a^7 γ κ^7 ϕs + 5 a^6 e^(2 a κ) (-1 + γ) κ^6 (-12 - 12 a κ + a^2 κ^2 + a^3 κ^3) ϕs ExpIntegralEi[-a κ]^2 +
     32 a^2 e^(2 a κ) κ^2 ExpIntegralEi[-2 a κ] (360 e^(a κ) β + (165 - 30 a κ - 150 a^2 κ^2 - 49 a^3 κ^3 - 4 a^4 κ^4 - 60 ϵst (-5 - 2 a κ + 3 a^2 κ^2) + γ (-45 + 90 a κ - 90 a^2 κ^2 + 61 a^3 κ^3 + 16 a^4 κ^4)) ϕs +
        45 a^4 e^(a κ) (-1 + γ) κ^4 ϕs ExpIntegralEi[-a κ]) + 11520 a^2 e^(3 a κ) β κ^2 Gamma[0, 2 a κ] + 5280 a^2 e^(2 a κ) κ^2 ϕs Gamma[0, 2 a κ] - 1440 a^2 e^(2 a κ) γ κ^2 ϕs Gamma[0, 2 a κ] +
     9600 a^2 e^(2 a κ) ϵst κ^2 ϕs Gamma[0, 2 a κ] + 960 a^3 e^(2 a κ) κ^3 ϕs Gamma[0, 2 a κ] + 2880 a^3 e^(2 a κ) γ κ^3 ϕs Gamma[0, 2 a κ] + 3840 a^3 e^(2 a κ) ϵst κ^3 ϕs Gamma[0, 2 a κ] - 2880 a^4 e^(2 a κ) κ^4 ϕs Gamma[0, 2 a κ] -
     2880 a^4 e^(2 a κ) γ κ^4 ϕs Gamma[0, 2 a κ] - 5760 a^4 e^(2 a κ) ϵst κ^4 ϕs Gamma[0, 2 a κ] - 1440 a^5 e^(2 a κ) κ^5 ϕs Gamma[0, 2 a κ] + 1440 a^5 e^(2 a κ) γ κ^5 ϕs Gamma[0, 2 a κ] +
     10 a^2 e^(a κ) κ^2 ExpIntegralEi[-a κ] ((1 + a κ) (12 e^(a κ) β (-12 + a^2 κ^2) + (-1 + γ) (84 - 12 a κ + 6 a^2 κ^2 - 10 a^3 κ^3 - a^4 κ^4 + a^5 κ^5) ϕs) + 144 a^4 e^(2 a κ) (-1 + γ) κ^4 ϕs Gamma[0, 2 a κ]))) //. {ϕs → ϕ0 * Exp[
     a κ], n0 → (κ^2) * ϵ0 * kB * T0 / (2 * el^2)}] + Sϵc;
```

In[176]:=
```
(* Large Debye length limit. Note that the functional dependencies are slightly
 different than for the splitting approach. This makes sense because of the limitations of this method. *)
```

In[177]:=
```
Normal[Series[Sc, {κ, 0, 2}, Assumptions → {a > 0, κ > 0}]] /. {(κ^2) → (2 * el^2) * n0 / (ϵ0 * kB * T0)}
```

Out[177]=

$$-\frac{\text{kB}^2\, \text{T0}^2\, \epsilon 0\, \epsilon\text{st}\, \phi 0^2}{105\, a\, \text{el}^2\, \text{eta}} - \frac{2\, \text{kB}^2\, \text{T0}^2\, \epsilon 0\, \epsilon\text{st}\, \kappa\, \phi 0^2}{105\, \text{el}^2\, \text{eta}} + \frac{a\, \text{kB n0 T0} \left(-420\, \beta\, \phi 0 - 49\, \phi 0^2 + 161\, \gamma\, \phi 0^2 + 16\, \epsilon\text{st}\, \phi 0^2\right)}{2520\, \text{eta}}$$

In[178]:=
```
(* Small Debye length limit. *)
```

In[179]:=
```
U1 = Limit[Sc[[1]], κ → Infinity, Assumptions → {a > 0, el > 0, eta > 0, kB > 0, T0 > 0, ϵ0 > 0, ϵst ∈ Reals, ϕ0 ∈ Reals}];
U2 = Limit[Sc[[2]], κ → Infinity, Assumptions → {a > 0, el > 0, eta > 0, kB > 0, T0 > 0, ϵ0 > 0, ϵst ∈ Reals, ϕ0 ∈ Reals, β ∈ Reals, γ ∈ Reals}];
Usmall = Simplify[U1 + U2]
```

Out[181]=

$$\frac{\text{kB}^2\, \text{T0}^2\, \epsilon 0\, \phi 0 \left(-8\, \beta + \phi 0 - \epsilon\text{st}\, \phi 0\right)}{12\, a\, \text{el}^2\, \text{eta}}$$

The Speed resulting from the Other Forces for an Insulating Surface

Last we integrate out the contributions for the insulating surface.

In[182]:=
```
(*** Sum the full series for the insulating surface ***)
```

In[183]:=
```
(* Uncomment the following lines to perform the calculation *)
(*exSs=Sum[IntTerm[k,exInts],{k,1,Length[exInts]}];*)
(*SsR=Kpref*Simplify[Expand[exSs]]*)
```

In[184]:=
```
(* The calculations are long so we present the result below. *)
```

In[185]:=
```
Ss = Simplify[(-1/(4320 eta (2 + 2 a κ + a^2 κ^2)) a e^(-2 a κ) kB n0 T0 ϕs (1440 e^(a κ) β - 960 a e^(a κ) β κ - 1920 a^2 e^(a κ) β κ^2 - 1200 a^3 e^(a κ) β κ^3 + 120 a^4 e^(a κ) β κ^4 + 120 a^5 e^(a κ) β κ^5 + 168 ϕs - 552 γ ϕs - 456 a κ ϕs + 744 a γ κ ϕs -
    2324 a^2 κ^2 ϕs + 1076 a^2 γ κ^2 ϕs - 2112 a^3 κ^3 ϕs + 528 a^3 γ κ^3 ϕs - 966 a^4 κ^4 ϕs + 294 a^4 γ κ^4 ϕs - 46 a^5 κ^5 ϕs + 334 a^5 γ κ^5 ϕs - 19 a^6 κ^6 ϕs + 211 a^6 γ κ^6 ϕs - 5 a^8 κ^8 ϕs + 5 a^8 γ κ^8 ϕs +
    32 a^2 e^(2 a κ) κ^2 (720 e^(a κ) β + (330 - 60 a κ - 135 a^2 κ^2 - 203 a^3 κ^3 - 8 a^4 κ^4 - 4 a^5 κ^5 - 60 ϵst (-13 - 7 a κ + a^2 κ^2 + 3 a^3 κ^3) + γ (-90 + 180 a κ - 45 a^2 κ^2 - 13 a^3 κ^3 + 32 a^4 κ^4 + 16 a^5 κ^5)) ϕs) ExpIntegralEi[-2 a κ] +
    10 a^2 e^(a κ) κ^2 (2 + 2 a κ + a^2 κ^2) (12 e^(a κ) β (-12 + a^2 κ^2) + (-1 + γ) (84 - 12 a κ + 6 a^2 κ^2 - 10 a^3 κ^3 - a^4 κ^4 + a^5 κ^5) ϕs) ExpIntegralEi[-a κ] +
    5 a^6 e^(2 a κ) (-1 + γ) κ^6 (-24 - 24 a κ - 10 a^2 κ^2 + 2 a^3 κ^3 + a^4 κ^4) ϕs ExpIntegralEi[-a κ]^2 + 23 040 a^2 e^(3 a κ) β κ^2 Gamma[0, 2 a κ] + 10 560 a^2 e^(2 a κ) κ^2 ϕs Gamma[0, 2 a κ] - 2880 a^2 e^(2 a κ) γ κ^2 ϕs Gamma[0, 2 a κ] +
    24 960 a^2 e^(2 a κ) ϵst κ^2 ϕs Gamma[0, 2 a κ] + 1920 a^3 e^(2 a κ) κ^3 ϕs Gamma[0, 2 a κ] + 5760 a^3 e^(2 a κ) γ κ^3 ϕs Gamma[0, 2 a κ] + 13 440 a^3 e^(2 a κ) ϵst κ^3 ϕs Gamma[0, 2 a κ] - 480 a^4 e^(2 a κ) κ^4 ϕs Gamma[0, 2 a κ] -
    1440 a^4 e^(2 a κ) γ κ^4 ϕs Gamma[0, 2 a κ] - 1920 a^4 e^(2 a κ) ϵst κ^4 ϕs Gamma[0, 2 a κ] - 4320 a^5 e^(2 a κ) κ^5 ϕs Gamma[0, 2 a κ] - 1440 a^5 e^(2 a κ) γ κ^5 ϕs Gamma[0, 2 a κ] - 5760 a^5 e^(2 a κ) ϵst κ^5 ϕs Gamma[0, 2 a κ])) //.
  {ϕs → a * el * σ0 * Exp[a * κ] / (kB * T0 * ϵ0 * (1 + κ * a)), n0 → κ^2 * ϵ0 * kB * T0 / (2 * el^2)}] + Sϵs;
```

In[186]:=
```
(* Large Debye Length Limit *)
```

In[187]:=
```
Normal[Series[Ss, {κ, 0, 2}, Assumptions → {a > 0, κ > 0}]] /. {(κ^2) → (2 * el^2) * n0 / (ϵ0 * kB * T0)}
```

Out[187]=

$$-\frac{a\,\epsilon st\,\sigma 0^2}{105\,eta\,\epsilon 0}+\frac{el\,n0\left(-420\,a^2\,kB\,T0\,\beta\,\epsilon 0\,\sigma 0-49\,a^3\,el\,\sigma 0^2+161\,a^3\,el\,\gamma\,\sigma 0^2+64\,a^3\,el\,\epsilon st\,\sigma 0^2\right)}{2520\,eta\,kB\,T0\,\epsilon 0^2}$$

In[188]:=
```
(* Small Debye Length Limit, which is vanishing as expected *)
```

In[189]:=
```
U1 = Limit[Ss[[1]], κ → Infinity, Assumptions → {a > 0, el > 0, eta > 0, kB > 0, T0 > 0, ϵ0 > 0, ϵst ∈ Reals, σ0 ∈ Reals}];
U2 = Limit[Ss[[2]], κ → Infinity, Assumptions → {a > 0, el > 0, eta > 0, kB > 0, T0 > 0, ϵ0 > 0, ϵst ∈ Reals, σ0 ∈ Reals, β ∈ Reals, γ ∈ Reals}];
Usmall = Simplify[U1 + U2] * ((2 * el^2) * n0 * λ0^2 / (ϵ0 * kB * T0))
```

Out[191]=
```
0
```